

NAG Library Routine Document

G02CHF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G02CHF performs a multiple linear regression with no constant on a set of variables whose sums of squares and cross-products about zero and correlation-like coefficients are given.

2 Specification

```

SUBROUTINE G02CHF(N, K1, K, SSPZ, LDSSPZ, RZ, LDRZ, RESULT, COEF,
1          LDCOE, RZNV, LDRZNV, CZ, LDCZ, WKZ, LDWKZ, IFAIL)
    INTEGER          N, K1, K, LDSSPZ, LDRZ, LDCOE, LDRZNV, LDCZ, LDWKZ,
1          IFAIL
    double precision    SSPZ(LDSSPZ,K1), RZ(LDRZ,K1), RESULT(13),
1          COEF(LDCOE,3), RZNV(LDRZNV,K), CZ(LDCZ,K),
2          WKZ(LDWKZ,K)

```

3 Description

G02CHF fits a curve of the form

$$y = b_1x_1 + b_2x_2 + \cdots + b_kx_k$$

to the data points

$$\begin{pmatrix} x_{11}, x_{21}, \dots, x_{k1}, y_1 \\ x_{12}, x_{22}, \dots, x_{k2}, y_2 \\ \vdots \\ x_{1n}, x_{2n}, \dots, x_{kn}, y_n \end{pmatrix}$$

such that

$$y_i = b_1x_{1i} + b_2x_{2i} + \cdots + b_kx_{ki} + e_i, \quad i = 1, 2, \dots, n.$$

The routine calculates the regression coefficients, $b_1, b_2, \dots, b_k$, (and various other statistical quantities) by minimizing

$$\sum_{i=1}^n e_i^2.$$

The actual data values $(x_{1i}, x_{2i}, \dots, x_{ki}, y_i)$ are not provided as input to the routine. Instead, input to the routine consists of:

- (i) The number of cases, n , on which the regression is based.
- (ii) The total number of variables, dependent and independent, in the regression, $(k+1)$.
- (iii) The number of independent variables in the regression, k .
- (iv) The $(k+1)$ by $(k+1)$ matrix $[\tilde{S}_{ij}]$ of sums of squares and cross-products about zero of all the variables in the regression; the terms involving the dependent variable, y , appear in the $(k+1)$ th row and column.
- (v) The $(k+1)$ by $(k+1)$ matrix $[\tilde{R}_{ij}]$ of correlation-like coefficients for all the variables in the regression; the correlations involving the dependent variable, y , appear in the $(k+1)$ th row and column.

The quantities calculated are:

- (a) The inverse of the k by k partition of the matrix of correlation-like coefficients, $[\tilde{R}_{ij}]$, involving only the independent variables. The inverse is obtained using an accurate method which assumes that this sub-matrix is positive-definite (see Section 8).

- (b) The modified matrix, $C = [c_{ij}]$, where

$$c_{ij} = \frac{\tilde{R}_{ij}\tilde{r}^{ij}}{\tilde{S}_{ij}}, \quad i, j = 1, 2, \dots, k,$$

where $\tilde{r}^{ij}$ is the (i, j) th element of the inverse matrix of $[\tilde{R}_{ij}]$ as described in (a) above. Each element of C is thus the corresponding element of the matrix of correlation-like coefficients multiplied by the corresponding element of the inverse of this matrix, divided by the corresponding element of the matrix of sums of squares and cross-products about zero.

- (c) The regression coefficients:

$$b_i = \sum_{j=1}^k c_{ij}\tilde{S}_{j(k+1)}, \quad i = 1, 2, \dots, k,$$

where $\tilde{S}_{j(k+1)}$ is the sum of cross-products about zero for the independent variable x_j and the dependent variable y .

- (d) The sum of squares attributable to the regression, SSR , the sum of squares of deviations about the regression, SSD , and the total sum of squares, SST :

$SST = \tilde{S}_{(k+1)(k+1)}$, the sum of squares about zero for the dependent variable, y ;

$$SSR = \sum_{j=1}^k b_j\tilde{S}_{j(k+1)}; \quad SSD = SST - SSR.$$

- (e) The degrees of freedom attributable to the regression, DFR , the degrees of freedom of deviations about the regression, DFD , and the total degrees of freedom, DFT :

$$DFR = k; \quad DFD = n - k; \quad DFT = n.$$

- (f) The mean square attributable to the regression, MSR , and the mean square of deviations about the regression, MSD :

$$MSR = SSR/DFR; \quad MSD = SSD/DFD.$$

- (g) The F value for the analysis of variance:

$$F = MSR/MSD.$$

- (h) The standard error estimate:

$$s = \sqrt{MSD}.$$

- (i) The coefficient of multiple correlation, R , the coefficient of multiple determination, R^2 , and the coefficient of multiple determination corrected for the degrees of freedom, $\bar{R}^2$:

$$R = \sqrt{1 - \frac{SSD}{SST}}; \quad R^2 = 1 - \frac{SSD}{SST}; \quad \bar{R}^2 = 1 - \frac{SSD \times DFT}{SST \times DFD}.$$

- (j) The standard error of the regression coefficients:

$$se(b_i) = \sqrt{MSD \times c_{ii}}, \quad i = 1, 2, \dots, k.$$

- (k) The t values for the regression coefficients:

$$t(b_i) = \frac{b_i}{se(b_i)}, \quad i = 1, 2, \dots, k.$$

4 References

Draper N R and Smith H (1985) *Applied Regression Analysis* (2nd Edition) Wiley

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the number of cases used in calculating the sums of squares and cross-products and correlation-like coefficients.

- 2: K1 – INTEGER *Input*
On entry: the total number of variables, independent and dependent ($k + 1$), in the regression.
Constraint: $2 \leq K1 \leq N$.

- 3: K – INTEGER *Input*
On entry: the number of independent variables k in the regression.
Constraint: $K = K1 - 1$.

- 4: SSPZ(LDSSPZ,K1) – *double precision* array *Input*
On entry: SSPZ(i, j) must be set to $\tilde{S}_{ij}$, the sum of cross-products about zero for the i th and j th variables, for $i, j = 1, 2, \dots, k + 1$; terms involving the dependent variable appear in row $k + 1$ and column $k + 1$.

- 5: LDSSPZ – INTEGER *Input*
On entry: the first dimension of the array SSPZ as declared in the (sub)program from which G02CHF is called.
Constraint: LDSSPZ $\geq$ K1.

- 6: RZ(LDRZ,K1) – *double precision* array *Input*
On entry: RZ(i, j) must be set to $\tilde{R}_{ij}$, the correlation-like coefficient for the i th and j th variables, for $i, j = 1, 2, \dots, k + 1$; coefficients involving the dependent variable appear in row $k + 1$ and column $k + 1$.

- 7: LDRZ – INTEGER *Input*
On entry: the first dimension of the array RZ as declared in the (sub)program from which G02CHF is called.
Constraint: LDRZ $\geq$ K1.

- 8: RESULT(13) – *double precision* array *Output*
On exit: the following information:
 RESULT(1) SSR , the sum of squares attributable to the regression;
 RESULT(2) DFR , the degrees of freedom attributable to the regression;
 RESULT(3) MSR , the mean square attributable to the regression;
 RESULT(4) F , the F value for the analysis of variance;
 RESULT(5) SSD , the sum of squares of deviations about the regression;
 RESULT(6) DFD , the degrees of freedom of deviations about the regression;
 RESULT(7) MSD , the mean square of deviations about the regression;
 RESULT(8) SST , the total sum of squares;
 RESULT(9) DFT , the total degrees of freedom;
 RESULT(10) s , the standard error estimate;
 RESULT(11) R , the coefficient of multiple correlation;
 RESULT(12) R^2 , the coefficient of multiple determination;

RESULT(13) $\bar{R}^2$, the coefficient of multiple determination corrected for the degrees of freedom.

- 9: COEF(LDCOE,3) – **double precision** array *Output*

On exit: for $i = 1, 2, \dots, k$, the following information:

COEF($i, 1$)

b_i , the regression coefficient for the i th variable.

COEF($i, 2$)

$se(b_i)$, the standard error of the regression coefficient for the i th variable.

COEF($i, 3$)

$t(b_i)$, the t value of the regression coefficient for the i th variable.

- 10: LDCOE – INTEGER *Input*

On entry: the first dimension of the array COEF as declared in the (sub)program from which G02CHF is called.

Constraint: LDCOE $\geq K$.

- 11: RZNV(LDRZNV,K) – **double precision** array *Output*

On exit: the inverse of the matrix of correlation-like coefficients for the independent variables; that is, the inverse of the matrix consisting of the first k rows and columns of RZ.

- 12: LDRZNV – INTEGER *Input*

On entry: the first dimension of the array RZNV as declared in the (sub)program from which G02CHF is called.

Constraint: LDRZNV $\geq K$.

- 13: CZ(LDCZ,K) – **double precision** array *Output*

On exit: the modified inverse matrix, C , where

$$CZ(i, j) = \frac{RZ(i, j) \times RZNV(i, j)}{SSPZ(i, j)}, \quad i, j = 1, 2, \dots, k.$$

- 14: LDCZ – INTEGER *Input*

On entry: the first dimension of the array CZ as declared in the (sub)program from which G02CHF is called.

Constraint: LDCZ $\geq K$.

- 15: WKZ(LDWKZ,K) – **double precision** array *Workspace*

- 16: LDWKZ – INTEGER *Input*

On entry: the first dimension of the array WKZ as declared in the (sub)program from which G02CHF is called.

Constraint: LDWKZ $\geq K$.

- 17: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the

recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $K1 < 2$.

$IFAIL = 2$

On entry, $K1 \neq (K + 1)$.

$IFAIL = 3$

On entry, $N < K1$.

$IFAIL = 4$

On entry, $LDSSPZ < K1$,
or $LDRZ < K1$,
or $LDCOEF < K$,
or $LDRZNV < K$,
or $LDCZ < K$,
or $LDWKZ < K$.

$IFAIL = 5$

This indicates that the k by k partition of the matrix RZ, which is to be inverted, is not positive-definite.

$IFAIL = 6$

This indicates that the refinement following the actual inversion fails, indicating that the k by k partition of the matrix RZ, which is to be inverted, is ill-conditioned. The use of G02DAF, which employs a different numerical technique, may avoid the difficulty.

$IFAIL = 7$

Unexpected error in F04ABF.

7 Accuracy

The accuracy of any regression routine is almost entirely dependent on the accuracy of the matrix inversion method used. In G02CHF, it is the matrix of correlation-like coefficients rather than that of the sums of squares and cross-products about zero that is inverted; this means that all terms in the matrix for inversion are of a similar order, and reduces the scope for computational error. For details on absolute accuracy, the relevant section of the document describing the inversion routine used, F04ABF, should be consulted. G02DAF uses a different method, based on F04AMF, and that routine may well prove more reliable numerically. It does not handle missing values, nor does it provide the same output as this routine.

If, in calculating F or any of the $t(b_i)$ (see Section 3), the numbers involved are such that the result would be outside the range of numbers which can be stored by the machine, then the answer is set to the largest quantity which can be stored as a **double precision** variable, by means of a call to X02ALF.

8 Further Comments

The time taken by G02CHF depends on k .

This routine assumes that the matrix of correlation-like coefficients for the independent variables in the regression is positive-definite; it fails if this is not the case.

This correlation matrix will in fact be positive-definite whenever the correlation-like matrix and the sums of squares and cross-products (about zero) matrix have been formed either without regard to missing values, or by eliminating **completely** any cases involving missing values for any variable. If, however, these matrices are formed by eliminating cases with missing values from only those calculations involving the variables for which the values are missing, no such statement can be made, and the correlation-like matrix may or may not be positive-definite. You should be aware of the possible dangers of using correlation matrices formed in this way (see the G02 Chapter Introduction), but if they nevertheless wish to carry out regressions using such matrices, this routine is capable of handling the inversion of such matrices, provided they are positive-definite.

If a matrix is positive-definite, its subsequent re-organisation by either of G02CEF or G02CFF will not affect this property and the new matrix can safely be used in this routine. Thus correlation matrices produced by any of G02BDF, G02BEF, G02BKF or G02BLF, even if subsequently modified by either G02CEF or G02CFF, can be handled by this routine.

It should be noted that the routine requires the dependent variable to be the last of the $k + 1$ variables whose statistics are provided as input to the routine. If this variable is not correctly positioned in the original data, the means, standard deviations, sums of squares and cross-products about zero, and correlation-like coefficients can be manipulated by using G02CEF or G02CFF to reorder the variables as necessary.

9 Example

This example reads in the sums of squares and cross-products about zero, and correlation-like coefficients for three variables. A multiple linear regression with no constant is then performed with the third and final variable as the dependent variable. Finally the results are printed.

9.1 Program Text

```
*      G02CHF Example Program Text
*      Mark 14 Revised. NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          K1, N, K, LDSSPZ, LDRZ, LDCOEF, LDRZNV, LDCZ,
+                     LDWKZ
      PARAMETER        (K1=3, N=5, K=K1-1, LDSSPZ=K1, LDRZ=K1, LDCOEF=K,
+                     LDRZNV=K, LDCZ=K, LDWKZ=K)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5, NOUT=6)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, J
*      .. Local Arrays ..
      DOUBLE PRECISION COEF(LDCOEF,3), CZ(LDCZ,K), RESULT(13),
+                     RZ(LDRZ,K1), RZNV(LDRZNV,K), SSPZ(LDSSPZ,K1),
+                     WKZ(LDWKZ,K)
*      .. External Subroutines ..
      EXTERNAL         G02CHF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G02CHF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) ((SSPZ(I,J), J=1, K1), I=1, K1),
+      ((RZ(I,J), J=1, K1), I=1, K1)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Sums of squares and cross-products about zero:'
      WRITE (NOUT,99999) (J, J=1, K1)
      WRITE (NOUT,99998) (I, (SSPZ(I,J), J=1, K1), I=1, K1)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Correlation-like coefficients:'
      WRITE (NOUT,99999) (J, J=1, K1)
```

```

WRITE (NOUT,99998) (I,(RZ(I,J),J=1,K1),I=1,K1)
WRITE (NOUT,*)
IFAIL = 1
*
CALL G02CHF(N,K1,K,SSPZ,LDSSPZ,RZ,LDRZ,RESULT,COEF,LDCOEF,RZNV,
+          LDRZNV,CZ,LDCZ,WKZ,LDWKZ,IFAIL)
*
IF (IFAIL.NE.0) THEN
  WRITE (NOUT,*)
  WRITE (NOUT,99997) ' ** G02CHF returned with IFAIL = ', IFAIL
ELSE
  WRITE (NOUT,*) 'Vble      Coef      Std err      t-value'
  WRITE (NOUT,99996) (I,(COEF(I,J),J=1,3),I=1,K)
  WRITE (NOUT,*)
  WRITE (NOUT,*) 'Analysis of regression table :-'
  WRITE (NOUT,*)
  WRITE (NOUT,*)
  +      Source      Sum of squares  D.F.      Mean square      F-val
+ue'
  WRITE (NOUT,*)
  WRITE (NOUT,99995) 'Due to regression', (RESULT(I),I=1,4)
  WRITE (NOUT,99995) 'About regression', (RESULT(I),I=5,7)
  WRITE (NOUT,99995) 'Total', (RESULT(I),I=8,9)
  WRITE (NOUT,*)
  WRITE (NOUT,99994) 'Standard error of estimate =', RESULT(10)
  WRITE (NOUT,99994) 'Multiple correlation (R) =', RESULT(11)
  WRITE (NOUT,99994) 'Determination (R squared) =', RESULT(12)
  WRITE (NOUT,99994) 'Corrected R squared =', RESULT(13)
  WRITE (NOUT,*)
  WRITE (NOUT,*)
  +      'Inverse of correlation matrix of independent variables:'
  WRITE (NOUT,99993) (J,J=1,K)
  WRITE (NOUT,99992) (I,(RZNV(I,J),J=1,K),I=1,K)
  WRITE (NOUT,*)
  WRITE (NOUT,*) 'Modified inverse matrix:'
  WRITE (NOUT,99993) (J,J=1,K)
  WRITE (NOUT,99992) (I,(CZ(I,J),J=1,K),I=1,K)
END IF
*
99999 FORMAT (1X,3I10)
99998 FORMAT (1X,I4,3F10.4)
99997 FORMAT (1X,A,I5)
99996 FORMAT (1X,I3,3F12.4)
99995 FORMAT (1X,A,F14.4,F8.0,2F14.4)
99994 FORMAT (1X,A,F8.4)
99993 FORMAT (1X,2I10)
99992 FORMAT (1X,I4,2F10.4)
END

```

9.2 Program Data

G02CHF Example Program Data

245.0000	99.0000	82.0000
99.0000	271.0000	52.0000
82.0000	52.0000	54.0000
1.0000	0.3842	0.7129
0.3842	1.0000	0.4299
0.7129	0.4299	1.0000

9.3 Program Results

G02CHF Example Program Results

Sums of squares and cross-products about zero:

	1	2	3
1	245.0000	99.0000	82.0000
2	99.0000	271.0000	52.0000
3	82.0000	52.0000	54.0000

Correlation-like coefficients:

	1	2	3
1	1.0000	0.3842	0.7129
2	0.3842	1.0000	0.4299
3	0.7129	0.4299	1.0000

Vble	Coef	Std err	t-value
1	0.3017	0.1998	1.5098
2	0.0817	0.1900	0.4299

Analysis of regression table :-

Source	Sum of squares	D.F.	Mean square	F-value
Due to regression	28.9857	2.	14.4929	1.7382
About regression	25.0143	3.	8.3381	
Total	54.0000	5.		

Standard error of estimate = 2.8876
 Multiple correlation (R) = 0.7326
 Determination (R squared) = 0.5368
 Corrected R squared = 0.2280

Inverse of correlation matrix of independent variables:

	1	2
1	1.1732	-0.4507
2	-0.4507	1.1732

Modified inverse matrix:

	1	2
1	0.0048	-0.0017
2	-0.0017	0.0043